Photon Sieve

The wavelength and focal length are taken to be (in meters)

In[380]:=
```
λ = 303.8 * 10^-10;
f = 2;
```

and the number of zones is taken to be

In[382]:=
```
numZ = 21;
```

Lets pick the packing factor (greater than 1) to be

In[383]:=
```
α = 1.5;
```

To get constructive interference at the focus, we need

In[384]:=
$$\texttt{zRad[n\_]} := \sqrt{n\,\lambda\,f + \frac{n^2\,\lambda^2}{4}}$$

The radius of each zone is then

In[385]:=
```
zR1 = Array[zRad, numZ]
zR2 = Array[zRad, numZ, 2];
```

Out[385]=
```
{0.000246495, 0.000348597, 0.000426943, 0.000492991, 0.000551181,
 0.000603788, 0.000652166, 0.000697194, 0.000739486, 0.000779487,
 0.000817533, 0.000853885, 0.000888752, 0.000922301, 0.000954673,
 0.000985982, 0.00101633, 0.00104579, 0.00107445, 0.00110236, 0.00112958}
```

We can then calculate the width of each zone

In[387]:=
```
zW = zR2 - zR1;
```

The radius of each hole in the photon sieve is then

In[388]:=
```
hR = zW / 2;
```

and the distance of the center of each hole from the center of the photon sieve is

In[389]:=
```
hC = zR1 + hR;
```

The circumference of a circle of radius *hC* is

In[390]:=
```
hCirc = 2 π hC;
```

The number of holes in each zone is then determined by the packing factor α

In[391]:=
$$\texttt{numH} = \texttt{Floor}\left[\frac{\texttt{hCirc}}{\texttt{zW}\,\alpha}\right];$$

The total number of holes is then

In[392]:=
```
Total[numH]
```

Out[392]=
```
2011
```

We can then find the angular location of each hole to be

In[393]:=
```
hAngFunc[n_] := Array[Identity, n]
```

In[394]:=
```
hAng = 2 π Map[hAngFunc, numH] / numH;
```

Build plot of the photon sieve

In[395]:=
```
holes = {};
```

In[396]:=
```
For[i = 1, i <= Length[hC], i += 2,
 For[j = 1, j <= Length[hAng[[i]]], j++,
  AppendTo[holes,
   Circle[{hC[[i]] Cos[hAng[[i, j]]], hC[[i]] Sin[hAng[[i, j]]]}, hR[[i]]]]
 ]
]
```

In[397]:=
```
Graphics[holes]
```

Out[397]=



The electric field diffracting through a circular hole in the Smythe-Kirchoff approximation is (Jackson 10.113) (assuming normal incidence)

In[398]:=
```
Eh = TransformedField["Spherical" → "Cartesian",
  (I E^(I k r))/r k a^2 E0 ({Sin[θ] Cos[ϕ], Sin[θ] Sin[ϕ], Cos[θ]} × {1, 0, 0})
  BesselJ[1, k a Sin[θ]]/(k a Sin[θ]), {r, θ, ϕ} → {x, y, z}]
```

Out[398]=

$$\left\{\frac{i\, a\, e^{i k \sqrt{x^2+y^2+z^2}}\, E0\, x\, z^2\, \mathrm{BesselJ}\left[1, \frac{a k \sqrt{x^2+y^2}}{\sqrt{x^2+y^2+z^2}}\right]}{(x^2+y^2)\,(x^2+y^2+z^2)} + \frac{i\, a\, e^{i k \sqrt{x^2+y^2+z^2}}\, E0\, y^2\, \mathrm{BesselJ}\left[1, \frac{a k \sqrt{x^2+y^2}}{\sqrt{x^2+y^2+z^2}}\right]}{(x^2+y^2)\,\sqrt{x^2+y^2+z^2}},\right.$$

$$\frac{i\, a\, e^{i k \sqrt{x^2+y^2+z^2}}\, E0\, y\, z^2\, \mathrm{BesselJ}\left[1, \frac{a k \sqrt{x^2+y^2}}{\sqrt{x^2+y^2+z^2}}\right]}{(x^2+y^2)\,(x^2+y^2+z^2)} - \frac{i\, a\, e^{i k \sqrt{x^2+y^2+z^2}}\, E0\, x\, y\, \mathrm{BesselJ}\left[1, \frac{a k \sqrt{x^2+y^2}}{\sqrt{x^2+y^2+z^2}}\right]}{(x^2+y^2)\,\sqrt{x^2+y^2+z^2}},$$

$$\left. -\frac{i\, a\, e^{i k \sqrt{x^2+y^2+z^2}}\, E0\, z\, \mathrm{BesselJ}\left[1, \frac{a k \sqrt{x^2+y^2}}{\sqrt{x^2+y^2+z^2}}\right]}{x^2+y^2+z^2}\right\}$$

The electric field of an offset hole is then

In[409]:=
```
Ehole[x0_, y0_, a_] = Eh /. {x → x - x0, y → y - y0};
```

In[400]:=
```
Eps = 0;
```

In[401]:=
```
For[i = 1, i <= Length[hC], i += 2,
 For[j = 1, j <= Length[hAng[[i]]], j++,
  Eps = Eps + Ehole[hC[[i]] Cos[hAng[[i, j]]], hC[[i]] Sin[hAng[[i, j]]], hR[[i]]]
 ]
]
```

In[402]:=
```
Eps = Eps /. {k → 2 π / λ, E0 → 1};
```

In[403]:=
```
Pps = Norm[Abs[Eps]];
```

In[404]:=

```
Plot[Pps /. {y → .00001, x → 0}, {z, 0, 3}]
```

Out[404]=



In[405]:=
```
DensityPlot[Pps /. y → .00001, {x, -.0005, .0005}, {z, 0.5, 3},
  PlotPoints → 100, ColorFunction -> "Rainbow", PlotRange → All]
```

Out[405]=



In[406]:=
```
DensityPlot3D[Pps, {x, -.0005, .0005}, {y, -.0005, .0005}, {z, 0.5, 3}]
```

Out[406]=



In[408]:=

In[408]:=